

Smoky letters

Linus Romer


Abstract

The wish for an individually designed thank-you card led to the idea of developing a “smoky” capital *D* that would look a bit different with each compilation but still elegant, such that it could be combined with a copperplate font to form the word *Danke* (German for *thanks*). This project was substantially facilitated by METAPOST and its random number generator.



From a single path to a bunch of paths


For a quick start, I read the coordinates of the most important points (mainly extrema) of the letter *D* in the *Calligra* font using the FontForge editor. These points were connected in a path like this:




```
beginfig(0);
u:=1mm; % standard unit
pickup pencircle scaled .2pt; % pen size
z1=(21u,34u); z2=(48u,52u); z3=(60u,42u);
z4=(26u,20u); z5=(8u,36u); z6=(52u,65u);
z7=(82u,40u); z8=(39u,0); z9=(6u,7u);
z10=(0,4u); z11=(13u,0); z12=(50u,13u);
z13=(68u,36u); z14=(91u,53u);
draw z1..tension 1.5 and 1..z2{right}..
z3{down}..tension 1.5 and 1..
z4{left}..z5{up}..tension 1.4 and 1..
z6{right}..z7{down}..z8{left}..
z9{left}..z10{down}...z11{right}
..z12..z13..z14;
endfig;
```

Then, a replacement for the `draw` macro was needed, namely `smokydraw`. This new macro shifts the main points of the path and constructs a new path with the same directions and tensions as the original path:

Linus Romer



In the final version of `smokydraw` the stem widths are additionally changed by using random numbers (`normaldeviate`).



```
def smokydraw expr p =
save widths,k,n,smokypath;
numeric widths,k,n;
k=length p;
n=10; % number of curves on each side
for j=0 upto k: % set random widths
widths[j]=5u*abs(normaldeviate);
endfor
path smokypath;
for s=1,-1: % both sides of p
for i=1 upto n: % curve index
smokypath:=
for j=0 upto k-1: % point index
(point j of p shifted
(dir(angle(direction j of p)+90)
*i/n*widths[j]*s)){direction j of p}
..tension posttension j of p
and pretension j+1 of p..
endfor
(point k of p shifted
(dir(angle(direction k of p)+90)
*i/n*widths[k]*s)){direction k of p};
draw smokypath;
endfor
endfor
draw p % original path (in the middle)
enddef;
```

`smokydraw` needs a predefined standard unit `u` and the macros `posttension` and `pretension` as described in *The METAFONTbook*. Enjoy!

◇ Linus Romer
Oberseestrasse 7, Schmerikon, 8716, Switzerland
linus.romer (at) gmx dot ch